Using the Median to Estimate a Parameter I am trying to use the Median to estimate the value for the parameter $a$ in the following PMF.
\begin{equation}
\label{eq1:givenpdf}
    \mathbb{P}\left[X=\frac{a}{n}\right] = \frac{36}{5}\frac{n^2}{\left(n+1\right)\left(n+2\right)\left(n+3\right)\left(n+4\right)}\ for\ n=1,2,\dots
\end{equation}
When equating the CDF to 0.5 to obtain a formula for the median, the following was obtained:
\begin{equation}
\label{eq:Median}
    \implies \frac{a(5a^2+9an+4n^2)}{5(a+2n)(a+3n)(a+4n)} = 0.5 
\end{equation}
Solving for $n$ yielded:
$$n = -0.3389a - (0.1687 \pm 0.2923i)a + (0.0590 \mp 0.1023i)a \approx (-0.5667 \pm 0.1900i)a$$
$$ \text{and}$$
$$ n= 0.0056 \left[-61a + 60.7433a + 21.2534a \right] \approx 0.1167a$$
For which I chose the real root, for obvious reasons.
How can I use this information to obtain an Estimator for $a$ using the median and apply it on the data set attached below (with is summary statistics)?
I tried researching online, however I am still getting confused because the median for the data set provided is $0.5$


 A: The median of this distribution is at $a/9$, since
$$P[X<\frac{a}9]=48.6\%$$
$$P[X=\frac{a}9]=\ \ 3.4\%$$
$$P[X>\frac{a}9]=48.0\%$$
So you can estimate $a$ as $9$ times the median.
A: (I've deleted my previous answer as it was a bit sloppy.)
Because in a comment you've mentioned the desire for other estimators, here is an implementation of a maximum likelihood estimator.  Note that this is not so straightforward because potential estimates of $a$ must result in $a/x$ (for any observed value of $X=x$) being a positive integer.
In short for large sample sizes the maximum likelihood estimator will be the maximum of the observed values.  For not-so-large sample sizes the maximum likelihood estimate can be found in the following manner.  Here I've used Mathematica but something similar can be produced in R.)  The example below generates data for $a=7.5$.
(* Generate a random sample of size 5 *)
p = Table[(36/5) n^2/((n + 1) (n + 2) (n + 3) (n + 4)), {n, 1, 10000}];
SeedRandom[12345]; 
nn = RandomChoice[p -> Range[1, 10000], 5]
x = 7.5/nn;  (* Here a = 7.5 *)

(* Find value for a that maximizes the log of the likelihood and 
results in (essentially) all integer values of a/x *)
maxLogL = -Infinity;
mlea =.;
Do[ahat = Max[x]*n; (* Potential estimate of a *)
 ss = (ahat/x - Round[ahat/x])^2 // Total;  (* Distance that ahat/x is from all integers *)
 (* Only attempt to update ahat if ahat/x consists of integers *)
 If[ss < 0.01,
  logL = Total[
    Log[36/5] + 2 Log[ahat] + 2 Log[#] - Log[ahat + #] - 
       Log[ahat + 2 #] - Log[ahat + 3 #] - Log[ahat + 4 #] & /@ x];
  If[logL > maxLogL, mlea = ahat; maxLogL = logL]
  ], {n, 1, 25}]

mlea
(* 7.5 *)

mlea/x  (* Check on mlea/x consists of all integers *)
(* {2, 5, 28, 7, 4} *)

Simulations result even for a sample size of 5 in a much, much better estimator than using 9 times the median.
SeedRandom[12345]; 
nSimulations = 10000;
mle = ConstantArray[0, nSimulations];
medianEstimator = ConstantArray[0, nSimulations];

Do[
 (* Generate a random sample of size 5 *)
 p = Table[(36/5) n^2/((n + 1) (n + 2) (n + 3) (n + 4)), {n, 1, 
    10000}];
 nn = RandomChoice[p -> Range[1, 10000], 5];
 x = 7.5/nn;
 
 (* Median estimator *)
 medianEstimator[[i]] = 9 Median[x];
 
 (* Find value for a that maximizes the log of the likelihood and 
 results in (essentially) all integer values of a/x *)
 maxLogL = -Infinity;
 mlea =.;
 Do[ahat = Max[x]*n; (* Potential estimate of a *)
  ss = (ahat/x - Round[ahat/x])^2 // Total;  (* Distance that ahat/
  x is from all integers *)
  (* Only attempt to update ahat if ahat/x consists of integers *)
  If[ss < 0.01,
   logL = 
    Total[Log[36/5] + 2 Log[ahat] + 2 Log[#] - Log[ahat + #] - 
        Log[ahat + 2 #] - Log[ahat + 3 #] - Log[ahat + 4 #] & /@ x];
   If[logL > maxLogL, mlea = ahat; maxLogL = logL]
   ], {n, 1, 25}];
 mle[[i]] = mlea,
 {i, 1, nSimulations}]

Here are the results:
{Mean[mle], StandardDeviation[mle], Median[mle]}
(* {7.37627, 0.697574, 7.5} *)

{Mean[medianEstimator], StandardDeviation[medianEstimator], Median[medianEstimator]}
(* {9.09117, 6.764, 7.5} *)

A plot of the relative frequencies of estimates for both the maximum likelihood and median estimators:

